



WDVV equations: symbolic computations of Hamiltonian operators

Jakub Vašíček¹ · Raffaele Vitolo^{2,3}

Received: 30 November 2021 / Revised: 17 May 2022 / Accepted: 31 May 2022 /
Published online: 14 July 2022
© The Author(s) 2022

Abstract

We describe software for symbolic computations that we developed in order to find Hamiltonian operators for Witten–Dijkgraaf–Verlinde–Verlinde (WDVV) equations, and verify their compatibility. The computation involves nonlocal (integro-differential) operators, for which specific canonical forms and algorithms have been used.

Keywords WDVV equations · Hamiltonian operators · Schouten bracket · Symbolic computations · Integro-differential operators · Weakly nonlocal operators

Mathematics Subject Classification 37K10 · 53D45 · 68W30

1 Introduction

The Witten–Dijkgraaf–Verlinde–Verlinde (WDVV) equations, also known as associativity equations, originated in $2D$ -topological field theory, but they acquired mathematical interest from the work of B. Dubrovin [6], who showed the deep connections between solutions of WDVV equations and integrable systems of PDEs, in particular, bi-Hamiltonian hierarchies.

Dedicated to the memory of Vladimir Gerdt (1947–2021).

✉ Raffaele Vitolo
raffaele.vitolo@unisalento.it
Jakub Vašíček
jakub.vasicek@math.slu.cz

¹ Mathematical Institute of the Silesian University, Silesian University, Opava, Czech Republic

² Department of Mathematics and Physics “E. De Giorgi”, University of Salento, Lecce, Italy

³ Sezione di Lecce, Istituto Nazionale di Fisica Nucleare, Lecce, Italy

Recently [25], we proved that, in low dimensions, the WDVV equations themselves are bi-Hamiltonian systems of PDEs. The proof required sophisticated symbolic computational tools that have been only recently made available [4, 5].

In this paper, we would like to describe the algorithms and symbolic software that we developed in order to carry out the computations in [25]. To do that, let us first describe the mathematical problem that we solved. Following [7], in \mathbb{R}^N we are to find a function $F = F(t^1, \dots, t^N)$ such that

1. $\frac{\partial^3 F}{\partial t^1 \partial t^\alpha \partial t^\beta} = \eta_{\alpha\beta}$ is a constant symmetric nondegenerate matrix;
2. $c_{\alpha\beta}^\gamma = \eta^{\gamma\epsilon} \frac{\partial^3 F}{\partial t^\epsilon \partial t^\alpha \partial t^\beta}$ are the structure constants of an associative algebra;
3. F is quasihomogeneous: $F(c^{d_1} t^1, \dots, c^{d_N} t^N) = c^{d_F} F(t^1, \dots, t^N)$.

The conditions of associativity, or WDVV equations, are the system of PDEs

$$\eta^{\mu\lambda} \frac{\partial^3 F}{\partial t^\lambda \partial t^\alpha \partial t^\beta} \frac{\partial^3 F}{\partial t^\nu \partial t^\mu \partial t^\gamma} = \eta^{\mu\lambda} \frac{\partial^3 F}{\partial t^\nu \partial t^\alpha \partial t^\mu} \frac{\partial^3 F}{\partial t^\lambda \partial t^\beta \partial t^\gamma}. \tag{1}$$

Item 1 implies that F can be rewritten as

$$F = \frac{1}{6} \eta_{11} (t^1)^3 + \frac{1}{2} \sum_{k>1} \eta_{1k} t^k (t^1)^2 + \frac{1}{2} \sum_{k,s>1} \eta_{sk} t^s t^k t^1 + f(t^2, \dots, t^N), \tag{2}$$

up to a second degree polynomial of the field variables. This implies that the WDVV system is an overdetermined system in one unknown function f of $N - 1$ independent variables.

A technique introduced in [17] (see also [10]) allows to rewrite the WDVV system as $N - 2$ commuting quasilinear first-order systems of conservation laws of the form

$$u_t^i = (V^i(\mathbf{u}))_x, \quad i = 1, \dots, n. \tag{3}$$

For example, if $N = 3$ and

$$\eta_{\alpha\beta} = \delta_{\alpha+\beta, N+1} = \begin{pmatrix} 0 & 0 & 1 \\ 0 & 1 & 0 \\ 1 & 0 & 0 \end{pmatrix}, \tag{4}$$

there is only one WDVV equation:

$$f_{ttt} = f_{xxt}^2 - f_{xxx} f_{xtt}. \tag{5}$$

Introducing the new coordinates $a = f_{xxx}$, $b = f_{xxt}$, $c = f_{xtt}$ we have the compatibility conditions

$$\begin{cases} a_t = b_x, \\ b_t = c_x, \\ c_t = (b^2 - ac)_x. \end{cases} \tag{6}$$

The above representation allowed to look for a bi-Hamiltonian formalism. That was found in [9]: the right-hand side of (6) was rewritten in two ways

$$u_t^i = A_1^{ij} \frac{\delta H_2}{\delta u^j} = A_2^{ij} \frac{\delta H_1}{\delta u^j}, \quad (7)$$

where $\delta/\delta u^i$ stands for the variational derivative, H_1 and H_2 are two Hamiltonian densities and A_1, A_2 are two compatible local Hamiltonian operators. Here, ‘compatible’ means that the pencil $A_1 + \lambda A_2$ is a Hamiltonian operator for every value of the parameter λ . We recall that the Hamiltonian property for a differential operator is equivalent to differential conditions on its coefficients, to be discussed later. Such a property is crucial for the integrability of the system of PDEs.

Only a few other examples of bi-Hamiltonian WDVV systems were known until recently. In [25] we were able to prove that, for $N = 3$, all WDVV systems admit a bi-Hamiltonian formulation, with Hamiltonian operators of the form

$$A_1^{ij} = g^{ij} \partial_x + \Gamma_k^{ij} u_x^k + \alpha V_q^i u_x^q \partial_x^{-1} V_p^j u_x^p + \beta \left(V_q^i u_x^q \partial_x^{-1} u_x^j + u_x^i \partial_x^{-1} V_q^j u_x^q \right) + \gamma u_x^i \partial_x^{-1} u_x^j, \quad (8)$$

$$A_2 = \partial_x (h^{ij} \partial_x + c_k^{ij} u_x^k) \partial_x, \quad (9)$$

with suitable Hamiltonian densities H_1 and H_2 . We observe that A_1 is a nonlocal first-order homogeneous Hamiltonian operator of Ferapontov type [8]. That is a member of a more general family of nonlocal operators, namely weakly nonlocal operators, that is discussed in [5]. In this context, ‘nonlocal’ means integro-differential operator (derivatives are taken with respect to the variable x). A_2 is a local third-order homogeneous Hamiltonian operator in canonical form [11, 12]. See more on these operators in next Section.

One of the computations that is needed in order to prove the bi-Hamiltonian property is the compatibility of the operators. That is made difficult by the fact that one of the operators is weakly-nonlocal (the Ferapontov operator A_1). The computation has been made possible by the recent computer algebra packages developed in [4].

Here we will describe how we used the Reduce computer algebra system, see <https://reduce-algebra.sourceforge.io/>, and its CDE package in particular (see [14] for a general description and [4] for calculations with weakly-nonlocal operators) in the nontrivial calculations that led to the results presented in [25]. We also checked some of the computations in Maple using the package `jacobi.mpl` from [4]. Maple was also used in a significant part of the computation in the case $N = 5$.

All files that are described in the text can be found in a GitHub repository. The files are also downloadable as a single `.zip` file, see [26].

2 Computing the first-order Hamiltonian operator A_1

Below we will focus on finding a first-order, weakly nonlocal Hamiltonian operator of Ferapontov type.

In the expression (8) (g^{ij}) is a non-degenerate matrix of functions of the field variables (u^i) , whose inverse (g_{ij}) can be interpreted as a covariant 2-tensor. The matrix V_q^i is just the Jacobian of the vector function (V^i) of the fluxes of the system (3), and α, β, γ are three constants.

The Hamiltonian property of A_1 is equivalent to the following conditions [8]:

- g^{ij} is symmetric;
- $\Gamma_{jk}^i = -g_{js}\Gamma_k^{si}$ are the Christoffel symbols of g_{ij} , regarded as a metric;
- the identities:

$$g^{ik}V_k^j = g^{jk}V_k^i, \tag{10a}$$

$$\nabla_k V_j^i = \nabla_j V_k^i, \tag{10b}$$

$$R_{kl}^{ij} = \alpha(V_k^i V_l^j - V_l^i V_k^j) + \beta(V_k^i \delta_l^j - V_k^j \delta_l^i - V_l^i \delta_k^j + V_l^j \delta_k^i) + \gamma(\delta_k^i \delta_l^j - \delta_l^i \delta_k^j) \tag{10c}$$

hold, where ∇ denotes the covariant derivative with respect to the Levi–Civita connection of g_{ij} and $R_{kl}^{ij} = g^{is}R_{skl}^j$ is the Riemannian curvature tensor. The above conditions also imply that the operator is a Hamiltonian operator for the system of PDEs (3) [24, 27].

Thus, finding an operator (8) actually reduces to finding the constants α, β and γ and the metric g^{ij} . However, finding the metric in general is by no means an easy computational task. To this end, we shall use a theorem from [2] which states that the metric of a first-order Hamiltonian operator for a non-diagonalizable quasilinear first-order system in $n = 3$ unknown functions is proportional to a contraction of the square of the Haantjes tensor:

$$g_{ij} = f H_{i\beta}^\alpha H_{j\alpha}^\beta = f H_{ij}, \quad f = f(\mathbf{u}). \tag{11}$$

We recall that the Haantjes tensor is obtained from (V^i) and its derivatives by means of a straightforward formula (see [25]).

After this simplification we are left with a much less demanding computational problem – we need to find one unknown function $f(\mathbf{u})$ and the above constants. To this end, we will use a computer algebra system to determine our unknowns.

2.1 The computation

The computation of a nonlocal first-order operator is divided into three smaller steps. Firstly, we will write down the candidate metric $g_{ij} = f H_{ij}$ and check (10a). Next we will solve the Eq. (10b) which will yield an explicit form of the metric g^{ij} and as a final step, we find the constants α, β and γ after plugging the found metric into the condition (10c). We will concentrate on a case from Mokhov-Pavlenko’s classification

[18], where the determining matrix η has the following form:

$$\eta^4 = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \lambda & 0 \\ 0 & 0 & \mu \end{pmatrix}, \quad \lambda^2 = 1, \quad \mu^2 = 1, \quad (12)$$

since it is not as computationally demanding and can be solved in full generality with respect to the constants λ, μ . The WDVV equation takes the form:

$$\mu f_{xxt}^2 - \mu f_{xxx} f_{xtt} + \lambda f_{xxt}^2 - \lambda f_{xxt} f_{ttt} - 1 = 0. \quad (13)$$

After introducing the coordinates $u^1 = f_{xxx}$, $u^2 = f_{xxt}$, $u^3 = f_{xtt}$ we obtain the first-order WDVV system in conservative form:

$$\begin{aligned} u_t^1 &= u_x^2, \\ u_t^2 &= u_x^3, \\ u_t^3 &= \left(\frac{\mu((u^2)^2 - u^1 u^3) + \lambda(u^3)^2 - 1}{\lambda u^2} \right)_x. \end{aligned} \quad (14)$$

All files used for computation are available at a [GitHub repository](#) [26].

2.1.1 The first step

The following computation can be found in the file `WDVV-3c-Eta4\dne3_1ho2.red`. In `Reduce`, we start by loading the package `cde` and initialization of our environment by the following sequence:

```
indep_var := {x};
dep_var := {u1, u2, u3};
total_order := 8;
cde({indep_var, dep_var, {}, total_order}, {});
```

followed by initialization of the right hand-side of the system:

```
de := {u2_x, u3_x,
      td((mu*u2^2 - mu*u1*u3 + lam*u3^2 - 1)/(lam*u2), x)};
```

where the last element is just the WDVV equation. Next, define the velocity matrix V_j^i (and its operator) of our system:

```
nc := length(dep_var);
matrix av(nc, nc);
for i:=1:nc do
  for j:=1:nc do
    av(i, j) := df(part(de, i), part(ford_var, j));
```

```
operator avt;
for i:=1:ncomp do for j:=1:ncomp do
  avt(i, j) := av(i, j);
```

We also need to load a `riemann4.red` library in order to generate the Nijenhuis and Haantjes tensors by the commands:

```
generate_all_nt (avt, dep_var);
generate_all_ht (avt, dep_var);
```

And finally the contraction of the square of Haantjes tensor is obtained by:

```
temphtc:={};
matrix hmet(nc,nc);
for i:=1:nc do
  for j:=1:nc do
    hmet(i,j):=
      <<
        temphtc:=for k:=1:nc join for h:=1:nc collect
          riemann_list2ids({ht_,k,i,h})*riemann_list2ids
            ({ht_,h,j,k});
        part(temphtc,0):=plus
      >>;
```

As a last step we check the symmetry condition $H_{ih}V_j^h = H_{jh}V_i^h$:

```
for i:=1:nc do for j:=i+1:nc do
  <<
    templhs:=(for h:=1:nc sum hmet(i,h)*av(h,j));
    temprhs:=(for h:=1:nc sum hmet(j,h)*av(h,i));
    write templhs - temprhs
  >>;
```

and save the result `hmet` for later computations. We remark that, by construction, `hmet` is a rational function of the field variables `u1`, `u2`, `u3` and the parameters `lam`, `mu`.

2.1.2 The second step

The following computation can be found in the file `WDVV-3c-Eta4\dne3_1ho3.red`. We, again, start by initializing the environment as in previous step. After constructing the velocity matrix V_j^i and loading the previous result we proceed to defining the metric g_{ij} as a functional factor of the contraction of the square of the Haantjes tensor:

```
for each el in dep_var do depend f,el;
gl1:=f*hmet;
```

Besides the metric itself, we will need its inverse g^{ij} and Christoffel symbols of the third kind Γ_k^{ij} (which are constructed inside the package `riemann3.red` by means of Christoffel symbols of the first and second kind):

```

gul:=g11^(-1);
operator g11_op,gul_op;
for i:=1:nc do for j:=1:nc do g11_op(i,j):=g11(i,j);
for i:=1:nc do for j:=1:nc do gul_op(i,j):=gul(i,j);

generate_all_chr1(g11_op,chr1_,dep_var);
generate_all_chr2(g11_op,gul_op,chr1_,chr2_,dep_var);
generate_all_chr3(g11_op,gul_op,chr2_,chr3_,dep_var);

```

```

operator gamma_hi;
for i:=1:nc do for j:=1:nc do for k:=1:nc do
  gamma_hi(i,j,k):=mk_chr3(chr3_,i,j,k);

```

To find the coefficients of the metric we will use the condition (10b) which is in our case equivalent to:

$$V_j^k \Gamma_k^{si} = V_k^s \Gamma_j^{ki}.$$

Hence, we first construct the left and right hand-side respectively:

```

operator ag1;
for all s,i,j let ag1(s,i,j)=
  (for k:=1:nc sum av(k,j)*gamma_hi(s,i,k));
operator ag2;
for all s,i,j let ag2(s,i,j)=
  (for k:=1:nc sum av(s,k)*gamma_hi(k,i,j));

```

and we assemble the system:

```

total_eq:=
  for i:=1:nc join
    for j:=1:nc join
      for s:=1:nc collect
        ag1(s,i,j) - ag2(s,i,j);

```

Using the package `crack`, a solver of overdetermined systems of PDEs [28, 29], we obtain a solution for our unknown function $f(\mathbf{u})$ by the following sequence:

```

split_vars:=cde_difflist(all_parametric_der,dep_var);
splitvars_total_eq:=splitvars_list(total_eq,split_vars);
unk:={f};

load_package crack;
lisp(max_gc_counter:=10000000000);
crack_results:=crack(splitvars_total_eq, {}, unk, {});

sol_unk:=second first crack_results;

```

Then by a mere substitution, have the explicit form of the metric g :

```
g11:=sub(sol_unk,g11);
gu1:=sub(sol_unk,gu1);
```

We again export this result in order to finish the final step - finding the nonlocal part of the first-order operator.

2.1.3 The third step

The following computation can be found in the file `WDVV-3c-Eta4\dne3_1ho4.red`. At last, we need to find the nonlocal tail of the operator A_1 . We simply need to plug everything we found into the condition (1) and see for which values of constants α, β, γ it is satisfied.

As usual, we initialize the environment. As an addition to that we load the results from the previous file and then construct the metric and generate all Christoffel symbols Γ_k^{ij} :

```
g11:=gu1^(-1);

operator g11_op,gu1_op;
for i:=1:nc do for j:=1:nc do g11_op(i,j):=g11(i,j);
for i:=1:nc do for j:=1:nc do gu1_op(i,j):=gu1(i,j);

generate_all_chr1(g11_op,chr1_,dep_var);
generate_all_chr2(g11_op,gu1_op,chr1_,chr2_,dep_var);
generate_all_chr3(g11_op,gu1_op,chr2_,chr3_,dep_var);
```

Let us also define the identity matrix which will represent the Kronecker δ_j^i :

```
idm:=mat((1,0,0),(0,1,0),(0,0,1));
```

and finally assemble the condition (10c):

```
eq_curv:=for i:=1:nc join
for j:=1:nc join
  for k:=1:nc join
    for h:=1:nc collect
      riem3(g11_op,gu1_op,chr2_,i,j,k,h,dep_var) -
      alp*(av(i,k)*av(j,h) - av(i,h)*av(j,k)) -
      bet*(av(i,k)*idm(j,h) - av(j,k)*idm(i,h) -
      av(i,h)*idm(j,k) + av(j,h)*idm(i,k)) -
      gam*(idm(i,k)*idm(j,h) - idm(i,h)*idm(j,k));
```

The system is a rational expression that should vanish for any value of the field variables u_1, u_2, u_3 . That yields an overdetermined linear system on the coefficients α, β, γ that can be solved only if $\alpha=\mu, \beta=0, \gamma=\lambda$. The substitution:


```
A:=sub({
  alp=mu,
  bet=0,
  gam=lam
},eq_curv);
write eq_curv;
```

yields a list of non-zero equations in terms of the parameters μ, λ . However, we should check only for all the possible combinations of the values of constants μ, λ :

```
sub({mu=1, lam=1}, A);
sub({mu=-1, lam=1}, A);
sub({mu=1, lam=-1}, A);
sub({mu=-1, lam=-1}, A);
```

which finally yields four lists of zeroes.

The metric g^{ij} that defines A_1 , according with the expression (8), is:

$$g^{ij} = \frac{\lambda}{\mu} \begin{pmatrix} -a^2\mu - b^2\lambda - 4\mu\lambda & -b(a\mu + c\lambda) & -b^2\mu - c^2\lambda - 1 \\ -b(a\mu + c\lambda) & -b^2\mu - c^2\lambda - 1 & \frac{c(ac\mu - 2b^2\mu - c^2\lambda + 1)}{b} \\ -b^2\mu - c^2\lambda - 1 & \frac{c(ac\mu - 2b^2\mu - c^2\lambda + 1)}{b} & \frac{\delta}{b^2} \end{pmatrix}, \quad (15)$$

where $\delta = 2ab^2c\lambda - a^2c^2\lambda + 2ac^3\mu - 2ac\mu\lambda - b^4\lambda - 3b^2c^2\mu - 2b^2\mu\lambda - c^4\lambda + 2c^2 - \lambda$ and the values of constants in (8) are $\alpha = \mu, \beta = 0, \gamma = \lambda$, with $\lambda, \mu = \pm 1$.

3 Computing the third-order Hamiltonian operator A_2

To find a bi-Hamiltonian structure for the system (3) we need a second Hamiltonian operator compatible with A_1 . It was proved in [25] that in the case $N = 3, 4$ and 5 the WDVV systems admit a third-order homogeneous Hamiltonian operator in the canonical form

$$A_2 = \partial_x (h^{ij} \partial_x + c_k^{ij} u_x^k) \partial_x \quad (16)$$

(we always require $\det(h^{ij}) \neq 0$; (h_{ij}) denotes the inverse matrix). We recall that the Hamiltonian property of A_2 is equivalent to the conditions

$$c_{ijk} = \frac{1}{3}(h_{ik,j} - h_{ij,k}). \quad (17a)$$

$$h_{mk,s} + h_{ks,m} + h_{ms,k} = 0, \quad (17b)$$

$$c_{msk,l} = -h^{pq} c_{pml} c_{qsk}. \quad (17c)$$

and that the Eq. (17b) implies that h_{ij} is a *Monge metric* [11, 12]. For computational purposes, that implies that the coefficients h_{ij} are quadratic polynomials of the field variables.

Then, it is also known [13] that the conditions under which a third-order Hamiltonian operator is the Hamiltonian operator for a quasi-linear system of first-order conservation laws are:

$$h_{im} V_j^m = h_{jm} V_i^m, \tag{18a}$$

$$c_{mkl} V_i^m + c_{mik} V_l^m + c_{mli} V_k^m = 0, \tag{18b}$$

$$h_{ks} V_{ij}^k = c_{smj} V_i^m + c_{smi} V_j^m. \tag{18c}$$

Thus, with respect to the problem of finding a third-order operator for WDVV systems, knowing the vector of fluxes ($V^i(\mathbf{u})$) reduces the above system of equations to a linear system of algebraic equations in the coefficients of the quadratic polynomials h_{ij} that can be solved in short time for small values of N .

For the particular example that we discussed in the previous section (system (14)) we can easily solve the system (18) and verify that the unique solution h_{ij} that we get fulfills the conditions (17b) and (17c). This implies that h_{ij} generates a third-order homogeneous Hamiltonian operator (16) using the formula (17a). We have the expression:

$$h_{ij} = \begin{pmatrix} b^2 + \mu & b\mu(\lambda c - \mu a) & -\mu\lambda b^2 \\ b\mu(\lambda c - \mu a) & \lambda + a^2 - \lambda c(2\mu a - \lambda c) & \lambda b(\mu a - \lambda c) \\ -\mu\lambda b^2 & \lambda b(\mu a - \lambda c) & b^2 \end{pmatrix}. \tag{19}$$

As this computation is not using any special package, there is no need to go into details in the code itself. The computation is provided in the file `WDVV-3c-Eta4\wdvv_3ord_op_eta4.red`

4 Compatibility of the Hamiltonian operators A_1, A_2

A *bi-Hamiltonian* system is just a system of PDEs which is Hamiltonian with respect to two Hamiltonian operators, A_1 and A_2 in our case, which are compatible. The compatibility condition can be either checked by means of the requirement that the pencil $A_1 + \lambda A_2$ is a Hamiltonian operator for every value of the parameter, or, equivalently, by computing the Schouten bracket $[A_1, A_2]$ (see [5]).

However, tensorial conditions which are equivalent to the compatibility condition of A_1 and A_2 as in our case are not available. So, the calculation should be made through the definition, which makes a pen-and-paper approach to the problem almost completely unfeasible. This is also due to the fact that an algorithm to do the calculation in the nonlocal case was missing until recently. A computational algorithm [5] and software packages [4] became available a short time ago. Through this software we checked the compatibility of the operators A_1 and A_2 .

All of the computations described below are contained in the files `wdvv_comp_n11_eta4.red` and `wdvv_comp_n12_eta4.red`. For convenience they are split into two separate files where in the first one we check the bracket $[A_1, A_1]$ and

then in the second file we check the compatibility condition $[A_1, A_2]$. Let us focus on the second computation.

We start as usual by initializing the environment:

```
indep_var := {x};
dep_var := {u1, u2, u3};
total_order := 8;
cde({indep_var, dep_var, {}, total_order}, {});

de := {u2_x, u3_x,
      td((mu*u2^2 - mu*u1*u3 + lam*u3^2 - 1)/(lam*u2), x)};
```

From now on, we initialize the Ferapontov operator A_1 . The metric of the first-order operator is loaded from another file:

```
in "dne3_lho3_res.red";
gul := gl1**(-1);
```

To construct the local part we further need to initialize the Christoffel symbols Γ_k^{ij} in the same way as before. We construct the local part of the operator by the following lines:

```
operator b;
for i:=1:nc do for j:=1:nc do
  b(i,j) := for k:=1:nc sum mk_chr3(chr3_, i, j, k)
    *part(dv1, k);

mk_cdifop(ham1_l, 1, {3}, 3);
for all i, j, psi let ham1_l(i, j, psi) =
  gul(i, j)*td(psi, x) + b(i, j)*psi;
```

where b represents the contraction $\Gamma_k^{ij}(\mathbf{u})u_x^k$.

The other, nonlocal, part of our operator consists of a nonlocal tail which is defined by the previously found values of constants α, β, γ :

```
mk_wnlop(c, w, 2);
c(1, 1) := mu;
c(2, 2) := rho;
c(1, 2) := 0;
c(2, 1) := 0;
for i:=1:nc do w(i, 1) := (for j:=1:nc sum av(i, j)
  *part(dv1, j));
w(1, 2) := u1_x;
w(2, 2) := u2_x;
w(3, 2) := u3_x;
```

Here the constants are represented by a symmetric matrix c^{ij} ; w_i^k on the other hand represents the nonlocal terms.

Now, we import the metric (in lower indices) of the third-order operator, stored in `gl3`, and we generate the constants $c_{ijk} = \frac{1}{3}(h_{ki,j} - h_{ji,k})$:

```
operator c_lo;
for i:=1:nc do for j:=1:nc do for k:=1:nc do
  c_lo(i,j,k) := (1/3)*(df(gl3(k,i),part(dep_var_equ,j))
    - df(gl3(j,i),part(dep_var_equ,k)));
```

then we raise two indices:

```
templist:={};
operator c_hi;
for i:=1:nc do for j:=1:nc do for k:=1:nc do
  c_hi(i,j,k) :=
  <<
    templist:=
      for m:=1:nc join for n:=1:nc collect
        gu3(n,i)*gu3(m,j)*c_lo(m,n,k);
    templist:=part(templist,0):=plus
  >>;
```

and finally we need the contraction $c_k^{ij} u_x^k$:

```
operator c_hi_con;
for i:=1:nc do for j:=1:nc do
  c_hi_con(i,j) :=
  <<
    templist:=for k:=1:nc collect
      c_hi(i,j,k)*mkid(part(dep_var_equ,k),!_x);
    templist:=part(templist,0):=plus
  >>;
```

We assemble the local part:

```
mk_cdiflop(ham2_l,1,{3},3);
for all i,j,psi let ham2_l(i,j,psi) =
td( gu3(i,j)*td(psi,x,2)+c_hi_con(i,j)*td(psi,x), x);
```

and even though there is no nonlocal part, we need to introduce a zero tail:

```
mk_wnlop(d,z,1);
d(1,1):=0;
for i:=1:nc do z(i,1):=0;
```

and put the two parts together:

```
ham2:={ham2_l,d,z};
```

All that remains to do is to configure the nonlocal variables. Here we have two distinct operators so we need three nonlocal variables and their names will be different this time:

```
nloc_var:={{tpsi,w,1},{tpsi,w,2},{tchi,z,1}};
nloc_arg:={{tpsi,w},{tchi,z}};
```

Finally we prepare the jet space

```
dep_var_tot:=cde_weaklyn1(
  indep_var,dep_var_equ,loc_arg,nloc_var,total_order);
```

and run the computation:

```
sb_res:=schouten_bracket_wnl(
  ham1,ham2,dep_var_equ,loc_arg,nloc_arg);
```

We have to plug in all the possible combinations of values $\mu, \lambda = \pm 1$ to obtain the desired result:

```
A:=sub({mu=1,rho=1},sb_res);
B:=sub({mu=-1,rho=1},sb_res);
C:=sub({mu=1,rho=-1},sb_res);
D:=sub({mu=-1,rho=-1},sb_res);
```

the result is always zero.

Thus we have proved by a direct computation that previously found operators A_1 and A_2 are compatible Hamiltonian operators.

5 A computation in Maple using `jacobi`

In this section we describe another tool which might be used to check if a weakly non-local operator is Hamiltonian and possibly compatible with another one: the package `jacobi` [4]. Since that package is written in Maple some readers might find it more convenient, and we thought that it is worth to explain how to achieve the same result with a different computer algebra system.

The main structural difference with the computation in the previous section is that `jacobi` uses the language of distributions, which is more popular among Theoretical Physicists.

The plan is the same as before, we need the metrics of both operators to generate their local parts and the three constants for the nonlocal part. However the notation is slightly different and that will be the main point of this section.

For the following computation we also need the external Maple library `jacobi.mpl` which can be found on [23]. The computation itself is contained in the file `WDVV-3c-Compat_Example_Eta4.mw`. For simplicity, we consider $\lambda = \mu = 1$ for this case.

We start by constructing A_1 . The metric is declared as follows:

```
g1 := Matrix(N, N);
g1[1, 1] := -u[1, x, 0]^2-u[2, x, 0]^2-4;
```

where by $u[1, x, 0]$ we denote in our case u_{0x}^1 (with the convention $u_{0x} = u$). Other coefficients are defined in a similar way. After the matrix, we need the Christoffel symbols which can be computed in Maple or, if we have them available from previous computations, input them as a three-dimensional field:

```

Gamma1 := Array(1 .. N, 1 .. N, 1 .. N);
Gamma1[1, 1, 1] := -u[1, x, 0];
Gamma1[1, 1, 2] := -u[2, x, 0];

```

and so on. Next, we input the two tail vectors $W1X$ and $W1Y$ which in this notation differ only by using a different symbol for the independent variable. The $W1X$ is introduced as follows:

```

W1X := Matrix(N, N)
W1X[1] := u[2, x, 1];
W1X[2] := u[3, x, 1];
W1X[3] := (-u[1, x, 0]*u[2, x, 0]*u[3, x, 1]+
  +u[1, x, 0]*u[2, x, 1]*u[3, x, 0]-
  -u[1, x, 1]*u[2, x, 0]*u[3, x, 0]+
  +u[2, x, 0]^2*u[2, x, 1]+2*u[2, x, 0]*u[3, x, 0]*u
    [3, x, 1]-
  -u[2, x, 1]*u[3, x, 0]^2+u[2, x, 1])/u[2, x, 0]^2;

```

and the same goes for $W1Y$ with y instead of x . The first-order nonlocal operator A_1 is then assembled by:

```

A1 := Array(1 .. N, 1 .. N, 0 .. M);
for i to N do
  for j to N do
    A1[i, j, 0] := g1[i, j]*delta[x-y, 1]+
      add(Gamma1[i, j, k]*u[k, x, 1], k = 1 .. N)*
      delta[x-y, 0]+W1X[i]*delta[x-y, -1]*W1Y[j]+
      +u[i, x, 1]*delta[x-y, -1]*u[j, y, 1]
    end do
  end do;

```

Next for the third-order operator we need its metric h_{ij} as well as constants c_k^{ij} . We input the metric, denoted by g_3 , as before and, similarly to Christoffel symbols for the previous metric, the constants c_k^{ij} are handled as a three-dimensional field:

```

c_hi := Array(1 .. N, 1 .. N, 1 .. N);
c_hi[2, 3, 1] := -1/u[2, x, 0];
c_hi[2, 3, 2] := (u[1, x, 0]-u[3, x, 0])/u[2, x, 0]^2;
c_hi[3, 3, 1] := 1/u[2, x, 0];

```

and so on. For the local operator the tail matrix is just a zero matrix:

```

W3 := Matrix(N, N)

```

Finally, the operator A_3 is assembled by:

```

A3 := Array(1 .. N, 1 .. N, 0 .. M)
for i to N do
  for j to N do

```

```

A3[i, j, 0] := g3[i, j]*delta[x-y, 3]
+add(
  (diff(g3[i, j], u[k, x, 0])+c_hi[i, j, k])
  *u[k, x, 1],
  k = 1 .. N)*delta[x-y, 2]
+add(
  add(
    (diff(c_hi[i, j, h], u[k, x, 0]))*u[k, x, 1]
    *u[h, x, 1],
    k = 1 .. N)+c_hi[i, j, h]*u[h, x, 2],
  h = 1 .. N)*delta[x-y, 1]
end do
end do;

```

We can then proceed to computing the Schouten bracket itself. We load the package `jacobi` by:

```
read `jacobi.mpl`;
```

The bracket is calculated by the following command and saved into an array `T3`:

```

T3 := Array(1 .. N, 1 .. N, 1 .. N);
Schouten_bracket(A3, A3, T3, N, M);

```

We can view the result just by printing the contents of `T3` (further simplification might be needed):

```

for i to N do
  for j to N do
    for k to N do
      print(T3[i, j, k])
    end do
  end do
end do;

```

The same command is used for computing the Schouten bracket of $[A_1, A_3]$, proving the compatibility of the two operators:

```

T13 := Array(1 .. N, 1 .. N, 1 .. N);
Schouten_bracket(P1, P3, T13, N, M);

```

6 Large scale computations

While doing the above computations for $N = 3$ is not a time demanding task, the situation quickly changes for $N = 4$, where just finding the third-order operator usually takes around 5 h on an average PC. To obtain the results for $N = 5$ in a similar way one would have to have an access to a huge amount of computational resources.

In this section, we describe the way in which we found the third-order operators in the case $N = 5$ (results in [25]). Since we did not even try to find a first-order operator

when $N = 5$, we will not deal with related problems. Only partial results are known in the case $N = 4$ [20].

There are two ways how to approach this problem. First, we could find a powerful enough machine to handle this computation. However, we are talking about hundreds of GB of RAM memory and possibly a week of computing time.

The other option is to optimize the computation itself, so that it not excessively resource-demanding. It can be done in a few ways, but the most effective at the beginning is to reduce the number of equations we try to solve at once.

As both approaches were used to obtain the results for $N = 5$ let us explore those below in more detail. All files used below are in a separate folder `\WDVV-5c-Large_scale`.

6.1 Using a supercomputer

If you have an access to a supercomputer it could be beneficial to try to use a rather “brute force” approach. Given the fact we cannot say for sure how many equations need to be solved in order to obtain the solution, it is by far the best first try.

To obtain the result in case of $N = 5$ for a canonical choice of η we have used the whole super-computing grid provided by a virtual organization *CESNET – Meta-Centrum*. The command to start the computation on the grid can be found in the file `starter.sh`.

It is also important to mention that we need to provide the file with the generated system in the conservative form as the algorithm has not been transferred into Maple yet. The algorithm itself is contained inside the file `w10_hydro_system_gen.red` whose output file (`w10_eta2_eq.red`) is then loaded into Maple. Finding the quasi-linear system in the conservative form using an algorithm was described in detail in [25] and is an interesting computational problem. However, since the program literally follows the steps described there, we will omit repeating the details. Note that some syntax adjustments are required before importing.

After we obtain the result from a remote computation it is important to have an independent check. We will use Reduce, file `w10_ham1_eta2_check.red`. We initialize as usual:

```
load_package cde;

nc:=10;

indep_var:={x};
dep_var:=for i:=1:10 collect mkid(a,i);
total_order:=6;
resname:="w10_ham1_eta_check_res.red";

cde({indep_var,dep_var,{},total_order},{});

in "w10_eta1_eq_transformed.red";
```



```
cap_v:=for each el in cons_laws_system collect second el;
in "w10_eta_gl3.red";
```

where in the files `w10_eta2_eq.red` and `w10_eta_gl3.red` we saved the general WDVV system and the result of the computation—the metric h_{ij} respectively which again needs to be transformed into the proper syntax.

Then we need to confirm that the Eq. (18) holds for the metric h_{ij} already in place (in form of the constants c_{ijk}) to confirm the compatibility with the first-order system. Lastly we need to check the Hamiltonian condition which we assemble by:

```
gu3:=gl3**(-1);
for k:=1:nc do
  for l:=1:nc do
    for m:=1:nc do
      for n:=1:nc do
        << if
          df(c_lo(m,n,k),part(dep_var,l)) + (
            for i:=1:nc sum (
              for j:=1:nc sum (
                gu3(i,j)*c_lo(i,m,l)*c_lo(j,n,k)
              )
            )
          ) neq 0 then rederr "Error - not Hamiltonian!"
        >>
```

which also passes in our case and confirms the result obtained by a super-computation.

6.2 Simplifying the computation

If we look into the algorithm we can quickly realize that it is massively inefficient in the sense of computational resources. The unknowns are exactly the coefficients of the quadratic polynomials in the metric of the third-order operator. In the case of $N = 5$ we need to find 1540 unknown constants.

On the other hand, the conditions (18) generate 2100 equations but those equations are polynomial with respect to the field variables. We need to collect the coefficients of each monomial out. There will be an immense amount of those collected equations. Just for comparison, in the case $N = 4$ we have 231 unknown coefficients, slightly more than 450 equations polynomial in the field variables and we generate around 350,000 of such linear equations.

It is now obvious that solving all of the collected equations is completely unnecessary and in this case a waste of resources. We could try to make an algorithm that will solve only a few of the equations which are polynomial in the field variables at once, plug the result into the rest and repeat the process batch by batch until all the equations are solved. This process could work, however, it is very hard to predict how much resources it will consume and what is the optimal size of the mentioned batch. The two extreme cases being the batch is too small and the number of iterations will

project into the time of the computation or the batch being too big literally clogging the memory and maybe never finish.

There is certainly enough space to optimize the whole process of solving those massively over-determined system of linear equations. However, by experiment we have found out that in our case there are around 60 equations out of the original 2100 polynomial ones needed to completely solve the posed problem. They originate from the system (20) below.

Also, saving partial results in the whole process is an absolute must. While the setup of the problem does not consume much time, relative to solving it, it is much easier to plug in a partial result from earlier and continue the computation.

More specifically, instead of solving the whole set of equations (18) we can simply restrict the system to

$$h_{ik} V_j^k = h_{jk} V_i^k, \quad i, j, k = 1, \dots, 4. \quad (20)$$

This computation takes around 8 hours on an average PC and the result is checked in the same way as in the previous case.

7 Conclusions

In this paper we described the computations that we performed in [25] in order to support the conjecture that all WDVV equations are endowed with bi-Hamiltonian formalism. These are symbolic software calculations that are nontrivial under both the viewpoint of the mathematical algorithm and the viewpoint of the symbolic programming that is needed to achieve the goal.

However, we think that further computational goals might be achieved to support and broaden the above conjecture.

- First of all, in the $N = 4$ case the bi-Hamiltonian property has been proved only for one of the two canonical forms of η , $\eta^{(1)}$. This was a result found in [20]. For the system arising from the second canonical form $\eta^{(2)}$ we found a third-order operator but we haven't even tried to find a first-order operator. It should be said that the method that we used in the case $N = 3$, coming from a characterization in [2], is not applicable in higher dimensions, hence the calculation might be a real challenge. Of course, the case $N = 5$ is even harder.

A way to find a first-order operator might be to restrict the search to matrices g^{ij} of rational functions of a certain degree. This common pattern can be observed in our examples (see also [25]) and might greatly simplify the search.

More generally, it should be possible to program a `cde` or `jacobi` computation of all first-order nonlocal operators that are compatible with a given third-order operator. This is a task for a future research project.

- Then, there is a well-known generalization of the WDVV equations, the oriented associativity equation [15]. These equations and the associated geometric structures are called F-manifolds with compatible flat structures [16], or simply flat F-manifolds. They share many properties with WDVV equations and Dubrovin-

Frobenius manifolds including the existence of an associated integrable dispersive hierarchy (see [1] for details).

We observe, in particular, that the oriented associativity equation has an infinite hierarchy of nonlocal symmetries [22], a first-order local Hamiltonian operator of the same type as A_1 [19] and a third-order *nonlocal* Hamiltonian operator which is the straightforward generalization of A_2 [3, 21].

Until now, the results on Hamiltonian operators are known only for one of the simplest cases of oriented associativity equation; more calculations are needed in order to support a conjecture on the bi-Hamiltonianity of the F-manifold equation. Indeed, the compatibility of A_1 and A_2 is still an open question even in the simplest case, see [21] for a discussion.

The above problems are on our schedule.

Acknowledgements R.V. would like to thank P. Lorenzoni and A. Sergyeyev for useful comments and suggestions. Computational resources were supplied by the project “e-Infrastruktura CZ” (e-INFRA CZ LM2018140) supported by the Ministry of Education, Youth and Sports of the Czech Republic. The research of J.V. was supported in part by the Specific Research grant SGS/13/2020 of the Silesian University in Opava, Czechia. The research of R.V. has been funded by the Dept. of Mathematics and Physics “E. De Giorgi” of the Università del Salento, Istituto Naz. di Fisica Nucleare IS-CSN4 *Mathematical Methods of Nonlinear Physics*, GNFM of Istituto Nazionale di Alta Matematica.

Funding Open access funding provided by Università del Salento within the CRUI-CARE Agreement.

Declarations

Conflict of interest The authors declare that they have no conflict of interest.

Open Access This article is licensed under a Creative Commons Attribution 4.0 International License, which permits use, sharing, adaptation, distribution and reproduction in any medium or format, as long as you give appropriate credit to the original author(s) and the source, provide a link to the Creative Commons licence, and indicate if changes were made. The images or other third party material in this article are included in the article’s Creative Commons licence, unless indicated otherwise in a credit line to the material. If material is not included in the article’s Creative Commons licence and your intended use is not permitted by statutory regulation or exceeds the permitted use, you will need to obtain permission directly from the copyright holder. To view a copy of this licence, visit <http://creativecommons.org/licenses/by/4.0/>.

References

1. Arsie, A., Buryak, A., Lorenzoni, P., Rossi, P.: Flat F-manifolds, F-CohFTs, and integrable hierarchies. *Commun. Math. Phys.* **1**, 2 (2021). <https://doi.org/10.1007/s00220-021-04109-8>
2. Bogoyavlenskij, O.I.: Necessary conditions for existence of non-degenerate Hamiltonian structures. *Commun. Math. Phys.* **182**, 253–290 (1996)
3. Casati, M., Ferapontov, E.V., Pavlov, M.V., Vitolo, R.F.: On a class of third-order nonlocal Hamiltonian operators. *J. Geom. Phys.* **138**, 285–296 (2019). <https://doi.org/10.1016/j.geomphys.2018.10.018>
4. Casati, M., Lorenzoni, P., Valeri, D., Vitolo, R.: Weakly nonlocal Poisson brackets: tools, examples, computations. (2021). [arXiv:2101.06467](https://arxiv.org/abs/2101.06467)
5. Casati, M., Lorenzoni, P., Vitolo, R.: Three computational approaches to weakly nonlocal poisson brackets. *Stud. Appl. Math.* **144**(4), 412–448 (2020). <https://doi.org/10.1111/sapm.12302>
6. Dubrovin, B.A.: Geometry of 2D topological field theories. In: *Integrable systems and quantum groups*, volume 1620 of *Lect. Notes Math.*, pp. 120–348. Springer, Berlin, Heidelberg (1996). [arXiv:org/abs/hep-th/9407018](https://arxiv.org/abs/hep-th/9407018)

7. Dubrovin, B.A.: *Encyclopedia of Mathematical Physics*, volume 1 A: A-C, chapter WDVV equations and Frobenius manifolds, p. 438–447. SISSA, Elsevier (2006). ISBN: 0125126611
8. Ferapontov, E.V.: Nonlocal Hamiltonian operators of hydrodynamic type: differential geometry and applications. *Am. Math. Soc. Transl.* **170**(2), 33–58 (1995)
9. Ferapontov, E.V., Galvao, C.A.P., Mokhov, O., Nutku, Y.: Bi-Hamiltonian structure of equations of associativity in 2-d topological field theory. *Commun. Math. Phys.* **186**, 649–669 (1997)
10. Ferapontov, E.V., Mokhov, O.I.: On the Hamiltonian representation of the associativity equations. In: Gelfand, I.M., Fokas, A.S., (eds) *Algebraic aspects of integrable systems: in memory of Irene Dorfman*, pp. 75–91. Birkhäuser Boston (1996)
11. Ferapontov, E.V., Pavlov, M.V., Vitolo, R.F.: Projective-geometric aspects of homogeneous third-order Hamiltonian operators. *J. Geom. Phys.* **85**, 16–28 (2014). <https://doi.org/10.1016/j.geomphys.2014.05.027>
12. Ferapontov, E.V., Pavlov, M.V., Vitolo, R.F.: Towards the classification of homogeneous third-order Hamiltonian operators. *Int. Math. Res. Not.* **22**, 6829–6855 (2016)
13. Ferapontov, E.V., Pavlov, M.V., Vitolo, R.F.: Systems of conservation laws with third-order Hamiltonian structures. *Lett. Math. Phys.* **108**(6), 1525–1550 (2018)
14. Krasil'shchik, J., Verbovetsky, A., Vitolo, R.: *The symbolic computation of integrability structures for partial differential equations*. Texts and Monographs in Symbolic Computation. Springer, 2018. ISBN 978-3-319-71654-1; see http://gdeq.org/Symbolic_Book for downloading program files that are discussed in the book
15. Losev, A., Manin, Y.I.: Extended modular operads. C. Hertling, M. Marcolli (Eds.), *Frobenius Manifolds, Quantum Cohomology and Singularities, Aspects of Math.*, vol. E36, Vieweg, Wiesbaden (2004), pp. 181–211
16. Manin, Y.: F-manifolds with flat structure and Dubrovin's duality. *Adv. Math.* **198**(1), 5–26 (2005)
17. Mokhov, O.I.: Symplectic and Poisson geometry on loop spaces of manifolds and nonlinear equations. In Novikov, S.P., (ed), *Topics in Topology and Mathematical Physics*, volume 170 of *Amer. Math. Soc. Transl. Ser. 2*, pp. 121–151. Amer. Math. Soc., Providence, R. I., 1995, [arXiv.org/abs/hep-th/9503076](https://arxiv.org/abs/hep-th/9503076)
18. Mokhov, O.I., Pavlenko, N.A.: Classification of the associativity equations with a first-order Hamiltonian operator. *Theor. Math. Phys.* **197**(1), 1501–1513 (2018)
19. Pavlov, M., Sergyeyev, A.: Oriented associativity equations and symmetry consistent conjugate curvilinear coordinate nets. *J. Geom. Phys.* **85**, 46–59 (2014)
20. Pavlov, M.V., Vitolo, R.F.: On the bi-Hamiltonian geometry of the WDVV equations. *Lett. Math. Phys.* **105**(8), 1135–1163 (2015)
21. Pavlov, M.V., Vitolo, R.F.: Bi-Hamiltonian structure of the Oriented Associativity equation. *J. Phys. A*, 52(20LT01) (2019). [arXiv:1812.01413](https://arxiv.org/abs/1812.01413). (published as a Letter)
22. Sergyeyev, A.: Infinite hierarchies of nonlocal symmetries of the Chen–Kontsevich–Schwarz type for the oriented associativity equations. *J. Phys. A* **42**, 404017 (2009)
23. The Geometry of Differential Equations website, webpage of the software packages and examples discussed in the paper [3]: https://gdeq.org/Weakly_nonlocal_Poisson_brackets (2021)
24. Tsarev, S.: On Poisson brackets and one-dimensional Hamiltonian systems of hydrodynamic type. *Soviet Math. Dokl.* **31**(3), 488–491 (1985)
25. Vašíček, J., Vitolo, R.: WDVV equations and invariant bi-Hamiltonian formalism. *J. High Energy Phys.*, 129 (2021). [arXiv:2101.13206](https://arxiv.org/abs/2101.13206)
26. Vašíček, J., Vitolo, R.: WDVV equations: Hamiltonian operators and symbolic computations. Computer algebra programs, 2021. GitHub: https://github.com/Jakub-Vasicek/WDVV-computations/tree/Computational-examples-bi-Hamiltonian/WDVV-Ham_op_and_symbolic_computations. Use the following link to retrieve all files in a single .zip file: <https://github.com/Jakub-Vasicek/WDVV-computations/archive/refs/heads/Computational-examples-bi-Hamiltonian.zip>
27. Vergallo, P., Vitolo, R.: Homogeneous Hamiltonian operators and the theory of coverings. *Differ. Geom. Appl.* **75**, 101713 (2021)
28. Wolf, T., Brand, A.: CRACK, user guide, examples and documentation. (2006) http://lie.math.brocku.ca/Crack_demo.html
29. Wolf, T., Brand, A.: Investigating DEs with CRACK and Related Programs, pp. 1–8. Special Issue, SIGSAM Bulletin (1995)